Aufgabe aus der Analysis

ÜBUNG [01]:
Beweisen Sie (mit den Mitteln der Integralrechnung): Für alle natürlichen Zahlen n gilt: $\sum_{n=1}^{N} (-1)^{n+1} \cdot \frac{1}{n} \cdot \binom{N}{n} = \sum_{m=1}^{N} \frac{1}{m}$

Jordan-Normalform

Aufgaben: 7, F1

```
> restart:
> with(LinearAlgebra):
```

In einem vorherigem Worksheet haben wir die Zerlegung in Haupträume betrachtet. Für diesen Abschnitt gehen wir davon aus, dass wir diese Zerlegung schon vorgenommen haben und betrachten daher den Fall, dass nur ein Hauptraum vorliegt.

Zyklische Haupträume

MATH: Sei A ein Endomorphismus des endlich erzeugten K-Vektorraumes V mit Minimalpolynom p^m und charakteristischem Polynom p^c mit $p \in K[x]$ normiert und irreduzibel (d.h. wir betrachten nur einen Hauptraum). Der Fall $m = c$ ist besonders ausgezeichnet: Wir haben einen bezüglich A **zyklischen Raum**, d. h. es existiert ein $v \in V$ mit $V = \langle v, Av, A^2v, \ldots \rangle$.

ÜBUNG [02]:
Sei $m = c$ wie eben. Gib eine Basis an, sodass der Endomorphismus durch die Begleitmatrix von p^c dargestellt wird.

BEIPSPIEL: (Zyklischer Raum, Vergleich mit Begleitmatrix)
Wir rechnen über <:

```
> p:=x^2+x+1;
```

$$p := x^2 + x + 1 \tag{2.1.1}$$

```
> irreduc(p);
```

$$\textit{true} \tag{2.1.2}$$

```
> A:=CompanionMatrix((x^2+x+1)^3,x);
```

$$A := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & -1 \\ 1 & 0 & 0 & 0 & 0 & -3 \\ 0 & 1 & 0 & 0 & 0 & -6 \\ 0 & 0 & 1 & 0 & 0 & -7 \\ 0 & 0 & 0 & 1 & 0 & -6 \\ 0 & 0 & 0 & 0 & 1 & -3 \end{bmatrix} \tag{2.1.3}$$

```
> map(i->(A^i).Column(IdentityMatrix(6),1), [$0..5]);
```

$$\left[\begin{bmatrix}1\\0\\0\\0\\0\\0\end{bmatrix}, \begin{bmatrix}0\\1\\0\\0\\0\\0\end{bmatrix}, \begin{bmatrix}0\\0\\1\\0\\0\\0\end{bmatrix}, \begin{bmatrix}0\\0\\0\\1\\0\\0\end{bmatrix}, \begin{bmatrix}0\\0\\0\\0\\1\\0\end{bmatrix}, \begin{bmatrix}0\\0\\0\\0\\0\\1\end{bmatrix}\right] \tag{2.1.4}$$

```
> factor(MinimalPolynomial(A,x));
  factor(CharacteristicPolynomial(A,x));
```

$$(x^2+x+1)^3$$

$$(x^2+x+1)^3 \tag{2.1.5}$$

ÜBUNG [03]:

Eine Matrix $B \in K^{n \times n}$ heißt **nilpotent**, falls ein $k \in \mathbb{N}$ existiert mit $B^k = 0$.
1.) Zeige, dass $k \le n$ gewählt werden kann.
2.) Gib für $n = 5$ eine nilpotente Matrix B und einen Vektor v an, so dass $K^{5 \times 1}$ bezüglich B zyklisch von v erzeugt wird.

MATH: Im vorliegenden **zyklischen** Fall sind die A-invarianten Teilräume linear durch Inklusion geordnet und gegeben durch $\mathrm{Bild}\left(p(A)^i\right) = \mathrm{Kern}\left(p(A)^{c-i}\right)$.

Im **BEISPIEL:**

```
> nu:=A^2+A+1;
  map(Rank, [IdentityMatrix(6), nu^1, nu^2, nu^3]);
```

$$v := \begin{bmatrix} 1 & 0 & 0 & 0 & -1 & 2 \\ 1 & 1 & 0 & 0 & -3 & 5 \\ 1 & 1 & 1 & 0 & -6 & 9 \\ 0 & 1 & 1 & 1 & -7 & 8 \\ 0 & 0 & 1 & 1 & -5 & 5 \\ 0 & 0 & 0 & 1 & -2 & 1 \end{bmatrix}$$

$$[6, 4, 2, 0] \tag{2.1.6}$$

Freiwillige ÜBUNG [F01]:

Gib ein Vertertersystem der Bahnen von $\mathrm{Aut}_A(V)$ auf V an.

MATH: Das folgende Programm erwartet ein irreduzibles Polynom p und eine natürliche Zahl a und gibt den zugehörigen (verallgemeinerten) **Jordan-Block** $J(p^a)$ aus:

```
> JBlock:=proc(p, x::name, m::posint,
  characteristic::nonnegint:=0)
   local M,d,i,irr:
   if not type(p, polynom(anything, x)) or degree(p,x) < 1
  then
    error sprintf("p ist kein Polynom in %a.", x):
   fi:
   if characteristic > 0 then
    if not isprime(characteristic) then
      error "Charakteristik muss eine Primzahl sein.";
     fi;
    irr := Irreduc(p) mod characteristic:
   else
    irr := irreduc(p):
   fi:
   if not irr then
    error "p ist nicht irreduzibel.":
   fi:
   d := degree(p, x):
   M := LinearAlgebra[DiagonalMatrix]([(LinearAlgebra
  [CompanionMatrix](p, x))$m]):
   for i from 1 to m-1 do
    M[d*i+1,d*i]:=1:
   od:
   M:
  end:
> JBlock(x+1, x, 4);
```

$$\begin{bmatrix} -1 & 0 & 0 & 0 \\ 1 & -1 & 0 & 0 \\ 0 & 1 & -1 & 0 \\ 0 & 0 & 1 & -1 \end{bmatrix} \quad \textbf{(2.1.7)}$$

Wir behandeln hier zusätzlich **verallgemeinerte Jordan-Blöcke**. So lässt sich ein Jordan-Block für eine Potenz eines irreduziblen Polynoms angeben, welches nicht Grad 1 hat:

```
> JBlock(x^2+x+1, x, 3);
```

$$\begin{bmatrix} 0 & -1 & 0 & 0 & 0 & 0 \\ 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & -1 & 0 & 0 \\ 0 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 0 & 1 & -1 \end{bmatrix} \quad \textbf{(2.1.8)}$$

Letzteres ist ein verallgemeinerter Jordan-Block für $K = \mathbb{Q}$ oder $K = \mathbb{R}$. Der verallgemeinerte Jordan-Block ist aber weiterhin nur für irreduzible Polynome definiert. Über $\mathbb{C}$ gibt es keinen Jordan-Block zu $x^2 + x + 1$, wohl aber zu seinen irreduziblen Faktoren:

```
> solve(x^2+x+1, x);
```

$$-\frac{1}{2} + \frac{1}{2} \mathrm{I}\sqrt{3}, -\frac{1}{2} - \frac{1}{2} \mathrm{I}\sqrt{3} \quad \textbf{(2.1.9)}$$

```
> JBlock(x-(-1/2+(1/2*I)*sqrt(3)), x, 3),JBlock(x-(-1/2-(1/2*
  I)*sqrt(3)), x, 3);
```

$$\begin{bmatrix} -\frac{1}{2} + \frac{1}{2} \mathrm{I}\sqrt{3} & 0 & 0 \\ 1 & -\frac{1}{2} + \frac{1}{2} \mathrm{I}\sqrt{3} & 0 \\ 0 & 1 & -\frac{1}{2} + \frac{1}{2} \mathrm{I}\sqrt{3} \end{bmatrix}, \quad \textbf{(2.1.10)}$$

$$\begin{bmatrix} -\frac{1}{2} - \frac{1}{2} \mathrm{I}\sqrt{3} & 0 & 0 \\ 1 & -\frac{1}{2} - \frac{1}{2} \mathrm{I}\sqrt{3} & 0 \\ 0 & 1 & -\frac{1}{2} - \frac{1}{2} \mathrm{I}\sqrt{3} \end{bmatrix}$$

Das bedeutet, dass wir über $\mathbb{C}$ eine Zerlegung in zwei Haupträume haben, während über $\mathbb{Q}$ oder $\mathbb{R}$ nur ein Hauptraum vorliegt. Weiteres Beispiel: $x^2 + 1$ ist irreduzibel über $\mathbb{Q}$, nicht aber über $\mathbb{F}_2$:

```
> JBlock(x^2+1, x, 2);
```

$$\begin{bmatrix} 0 & -1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & -1 \\ 0 & 0 & 1 & 0 \end{bmatrix} \tag{2.1.11}$$

```
> JBlock(x^2+1, x, 2, 2);
Error, (in JBlock) p ist nicht irreduzibel.
```

Der verallgemeinerte Jordanblock ermöglicht es uns, eine (verallgemeinerte) Jordan-Normalform zu definieren, die nicht erfordert, dass K algebraisch abgeschlossen ist.

MATH: Im zyklischen Fall kann man auf eine verallgemeinerte Jordan-Block-Matrix konjugieren.

MAPLE hat eine etwas andere Auffassung davon, was eine solche Jordan-Block-Matrix ist und liefert auch nur etwas, wenn p vom Grad 1, ist:

```
> JordanBlockMatrix([[-1,4]]);
```

$$\begin{bmatrix} -1 & 1 & 0 & 0 \\ 0 & -1 & 1 & 0 \\ 0 & 0 & -1 & 1 \\ 0 & 0 & 0 & -1 \end{bmatrix} \tag{2.1.12}$$

Wir hätten diese Matrix transponiert, weil wir eine andere Konvention benutzen, die durch den nilpotenten Fall von oben motiviert ist. (Überprüfe die Konvention in deiner Vorlesung.)

Was müssen wir im vorliegenden Fall A mit unserem p tun, um die Basis zu erhalten, die A auf ihren verallgemeinerten Jordan-Block konjugiert?
Wir gehen von der nilpotenten Matrix $\nu = p(A)$ aus:

```
> p;
```

$$x^2 + x + 1 \tag{2.1.13}$$

```
> F:=[1,x,p,x*p,p^2,x*p^2];
```

$$F := \left[1, x, x^2 + x + 1, x\,(x^2 + x + 1), (x^2 + x + 1)^2, x\,(x^2 + x + 1)^2\right] \tag{2.1.14}$$

DENKANSTOSS: Warum ist man versucht, dies die Nilpotenz-Variante der zyklischen Basis

```
> map(i->x^i,[$0..5]);
```

$$\left[1, x, x^2, x^3, x^4, x^5\right] \tag{2.1.15}$$

zu nennen? Was würde man bei kleinerem/größerem Grad von p tun?

Es ist $\mathrm{Grad}(p) = 2$ und $c = 3$. Wir haben F aufgestellt als $[x^0, \ldots, x^{\mathrm{Grad}(p) - 1},$

$x^0 p, \ldots, x^{\mathrm{Grad}(p)-1} p, \ldots, x^0 p^{c-1}, \ldots, x^{\mathrm{Grad}(p)-1} p^{c-1}]$ (wie sieht dies im algebraisch abgeschlossenen Fall aus, also für $\mathrm{Grad}(p) = 1$?). Nun setzen wir $x = A$ ein:

```
> F:=[IdentityMatrix(6),A,nu,A.nu,nu^2,A.nu^2];
```

$$F := \left[\begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & -1 \\ 1 & 0 & 0 & 0 & 0 & -3 \\ 0 & 1 & 0 & 0 & 0 & -6 \\ 0 & 0 & 1 & 0 & 0 & -7 \\ 0 & 0 & 0 & 1 & 0 & -6 \\ 0 & 0 & 0 & 0 & 1 & -3 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 & 0 & -1 & 2 \\ 1 & 1 & 0 & 0 & -3 & 5 \\ 1 & 1 & 1 & 0 & -6 & 9 \\ 0 & 1 & 1 & 1 & -7 & 8 \\ 0 & 0 & 1 & 1 & -5 & 5 \\ 0 & 0 & 0 & 1 & -2 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & -1 & 2 & -1 \\ 1 & 0 & 0 & -3 & 5 & -1 \\ 1 & 1 & 0 & -6 & 9 & -1 \\ 1 & 1 & 1 & -7 & 8 & 2 \\ 0 & 1 & 1 & -5 & 5 & 2 \\ 0 & 0 & 1 & -2 & 1 & 2 \end{bmatrix}, \begin{bmatrix} 1 & 0 & -1 & 1 & 0 & -1 \\ 2 & 1 & -3 & 2 & 1 & -3 \\ 3 & 2 & -5 & 3 & 2 & -5 \\ 2 & 3 & -5 & 2 & 3 & -5 \\ 1 & 2 & -3 & 1 & 2 & -3 \\ 0 & 1 & -1 & 0 & 1 & -1 \end{bmatrix}, \begin{bmatrix} 0 & -1 & 1 & 0 & -1 & 1 \\ 1 & -3 & 2 & 1 & -3 & 2 \\ 2 & -5 & 3 & 2 & -5 & 3 \\ 3 & -5 & 2 & 3 & -5 & 2 \\ 2 & -3 & 1 & 2 & -3 & 1 \\ 1 & -1 & 0 & 1 & -1 & 0 \end{bmatrix} \right] \tag{2.1.16}$$

Dies liefert uns die richtige Basis:

```
> map(i->Column(i,1),F);
```

$$\left[\begin{bmatrix} 1 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 1 \\ 1 \\ 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 1 \\ 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 1 \\ 2 \\ 3 \\ 2 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 2 \\ 3 \\ 2 \\ 1 \end{bmatrix} \right] \tag{2.1.17}$$

```
> T:=Matrix(%);
```

$$T := \begin{bmatrix} 1 & 0 & 1 & 0 & 1 & 0 \\ 0 & 1 & 1 & 1 & 2 & 1 \\ 0 & 0 & 1 & 1 & 3 & 2 \\ 0 & 0 & 0 & 1 & 2 & 3 \\ 0 & 0 & 0 & 0 & 1 & 2 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \tag{2.1.18}$$

```
> T^(-1).A.T,JBlock(p,x,3);
```

$$\begin{bmatrix} 0 & -1 & 0 & 0 & 0 & 0 \\ 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & -1 & 0 & 0 \\ 0 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 0 & 1 & -1 \end{bmatrix}, \begin{bmatrix} 0 & -1 & 0 & 0 & 0 & 0 \\ 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & -1 & 0 & 0 \\ 0 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & 0 & 1 & -1 \end{bmatrix} \quad \textbf{(2.1.19)}$$

Wichtig war hierbei, dass die gewählte Spalte aus

```
> F[1];
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \quad \textbf{(2.1.20)}$$

nicht im Spaltenraum von

```
> F[3];
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & -1 & 2 \\ 1 & 1 & 0 & 0 & -3 & 5 \\ 1 & 1 & 1 & 0 & -6 & 9 \\ 0 & 1 & 1 & 1 & -7 & 8 \\ 0 & 0 & 1 & 1 & -5 & 5 \\ 0 & 0 & 0 & 1 & -2 & 1 \end{bmatrix} \quad \textbf{(2.1.21)}$$

lag und ebenso die gewählte Spalte von F_3 nicht im Spaltenraum von

```
> F[5];
```

$$\begin{bmatrix} 1 & 0 & -1 & 1 & 0 & -1 \\ 2 & 1 & -3 & 2 & 1 & -3 \\ 3 & 2 & -5 & 3 & 2 & -5 \\ 2 & 3 & -5 & 2 & 3 & -5 \\ 1 & 2 & -3 & 1 & 2 & -3 \\ 0 & 1 & -1 & 0 & 1 & -1 \end{bmatrix} \quad \textbf{(2.1.22)}$$

lag. Wir haben immer die erste Spalte genommen, was auch meistens gutgeht.

ÜBUNG [04]:

Man gebe die Matrix an, die die Begleitmatrix von $(x-1)^5$ auf ihren Jordanblock transformiert.

```
> B:=CompanionMatrix((x-1)^5,x);
```

$$B := \begin{bmatrix} 0 & 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 & -5 \\ 0 & 1 & 0 & 0 & 10 \\ 0 & 0 & 1 & 0 & -10 \\ 0 & 0 & 0 & 1 & 5 \end{bmatrix} \tag{2.1.23}$$

(Hinweis: Wende das eben vorgeführte Verfahren an. Warum ist es in diesem Fall einfacher?)

Zerlegung eines Hauptraumes in zyklische Räume

MATH: Ist V nicht mehr zyklisch bezüglich A, aber immer noch mit Minimalpolynom p^m und charakteristischen Polynom p^c, wo $p \in K[x]$ normiert und irreduzibel ist ($m < c$), so kann man V in eine direkte Summe bezüglich A zyklischer Teilräume zerlegen. Z. B.

$$V := \langle v_1, Av_1, A^2v_1, \ldots \rangle \oplus \langle v_2, Av_2, A^2v_2, \ldots \rangle \oplus \ldots = \langle v_1 \rangle_A \oplus \langle v_2 \rangle_A \oplus \ldots$$

Dass V Summe solcher zyklischer Teilräume ist, ist wegen der endlichen Dimension klar (**DENKANSTOSS**: Warum?). Unsere Aufgabe besteht also darin, von der Summe zur direkten Summe überzugehen.

BEISPIEL:

```
> p := x^2+x+1;
```

$$p := x^2 + x + 1 \tag{2.2.1}$$

```
> A:=DiagonalMatrix([CompanionMatrix(p^3,x),CompanionMatrix(p,
  x)]):
> T:=SubMatrix(Transpose(A),1..8,[8,7,6,5,4,3,2,1])^2:
> A:=T.A.T^(-1);
```

(2.2.2)

$$A := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 1 & 1 \\ -6 & -22 & -39 & -49 & -43 & -3 & -109 & -96 \\ 2 & 6 & 7 & 6 & 3 & 1 & 6 & 6 \\ 0 & 0 & 1 & 0 & 0 & 0 & 7 & 7 \\ 0 & 0 & 0 & 1 & 0 & 0 & 6 & 6 \\ 23 & 90 & 179 & 249 & 237 & 11 & 553 & 475 \\ 0 & 1 & 3 & 6 & 7 & 0 & -3 & -6 \\ 0 & -1 & -3 & -6 & -8 & 0 & 0 & 3 \end{bmatrix} \qquad \textbf{(2.2.2)}$$

Wir wissen dass p^3 das Minimalpolynom von A ist:

```
> eval(subs(x=A,p^3)),eval(subs(x=A,p^2));
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}, \qquad \textbf{(2.2.3)}$$

$$\begin{bmatrix} 1 & 0 & -1 & 1 & 0 & 0 & 1 & 1 \\ -116 & -264 & -236 & -292 & -176 & -44 & -72 & -72 \\ 7 & 12 & 9 & 15 & 8 & 2 & 5 & 5 \\ 8 & 18 & 16 & 20 & 12 & 3 & 5 & 5 \\ 5 & 12 & 11 & 13 & 8 & 2 & 3 & 3 \\ 586 & 1362 & 1230 & 1494 & 908 & 227 & 359 & 359 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -2 & -6 & -6 & -6 & -4 & -1 & -1 & -1 \end{bmatrix}$$

Wir haben wieder die nilpotente Matrix

```
> P:=eval(subs(x=A,p)):
  P,P^3;
```

(2.2.4)

$$\left[\begin{array}{cccccccc} 1 & 0 & 0 & 0 & -1 & 0 & -2 & -2 \\ -21 & -54 & -79 & -73 & 80 & -9 & 116 & 116 \\ 3 & 6 & 8 & 6 & -3 & 1 & -9 & -9 \\ 2 & 6 & 8 & 7 & -4 & 1 & -8 & -8 \\ 0 & 0 & 1 & 1 & -5 & 0 & -5 & -5 \\ 94 & 252 & 374 & 357 & -418 & 42 & -571 & -571 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -1 & 2 & 0 & 1 & 1 \end{array}\right], \left[\begin{array}{cccccccc} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{array}\right] \quad \textbf{(2.2.4)}$$

```
> E1:=Column(IdentityMatrix(8),1):
> T1:=Matrix(map(i->(A^i).E1,[$0..5])): Rank(T1);
  Matrix(map(i->(A^i).E1,[$0..6])): Rank(%);
```

6

6 **(2.2.5)**

Also hat der von E_1 erzeugte A-zyklische Teilraum $\langle E_1\rangle_A$ die Dimension 6.
Ebenso der von E_2 erzeugte:

```
> E2:=Column(IdentityMatrix(8),2):
> T2:=Matrix(map(i->(A^i).E2,[$0..5])): Rank(T2);
  Matrix(map(i->(A^i).E2,[$0..6])): Rank(%);
```

6

6 **(2.2.6)**

```
> Rank(<T1|T2>);
```

8 **(2.2.7)**

Also ist klar: Die Summe der von E_1 und von E_2 erzeugten zyklischen Teilräume von V ist ganz V. Auch ist klar, dass der Schnitt 4-dimensional ist:

1.) $\langle E_1\rangle_A + \langle E_2\rangle_A = V$

2.) $\dim\left(\langle E_1\rangle_A \cap \langle E_2\rangle_A\right) = 4.$

Also nach unseren Einsichten über zyklische Räume für A:

$\langle E_1\rangle_A \cap \langle E_2\rangle_A = P\left(\langle E_1\rangle_A\right)$

Wir versuchen, E_2 durch einen anderen Vektor zu ersetzen, wo der Schnitt kleiner, aber das Gesamterzeugtnis immer noch ganz V ist.
Da $P^3E_2 = 0$ aber

```
> (P^2).E2;
```

(2.2.8)

$$\begin{bmatrix} 0 \\ -264 \\ 12 \\ 18 \\ 12 \\ 1362 \\ 0 \\ -6 \end{bmatrix} \tag{2.2.8}$$

noch ungleich Null, versuchen wir E_2 so zu modifizieren, dass wir Null bekommen. Also versuchen wir $P^2 E_2$ linear zu kombinieren aus $P^2(E_1)$ und $P^2(AE_1)$, die eine K-Basis des minimalen A-invarianten Teilraumes von $\langle E_1 \rangle_A$ sind:

```
> Test1:=<(P^2).E1|(P^2).A.E1|(P^2).E2>;
```

$$Test1 := \begin{bmatrix} 1 & -2 & 0 \\ -116 & 100 & -264 \\ 7 & -8 & 12 \\ 8 & -7 & 18 \\ 5 & -4 & 12 \\ 586 & -491 & 1362 \\ 0 & 0 & 0 \\ -2 & 1 & -6 \end{bmatrix} \tag{2.2.9}$$

```
> nn:=op(NullSpace(Test1));
```

$$nn := \begin{bmatrix} -4 \\ -2 \\ 1 \end{bmatrix} \tag{2.2.10}$$

Wir ersetzen E_2 durch

```
> E2n:=<E1|A.E1|E2>.nn;
```

(2.2.11)

$$E2n := \begin{bmatrix} -4 \\ 13 \\ -4 \\ 0 \\ 0 \\ -46 \\ 0 \\ 0 \end{bmatrix} \quad \textbf{(2.2.11)}$$

DENKANSTOSS: Warum ist immer noch $\langle E_1, E_{2n}\rangle_A = V$?

```
> (P^2).E2n,P.E2n;
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} -4 \\ 112 \\ -12 \\ -8 \\ -4 \\ -528 \\ 0 \\ 0 \end{bmatrix} \quad \textbf{(2.2.12)}$$

Wir haben also einen Forschritt: E_{2n} hat Länge 2 statt 3, d. h. es wird nach 2-maligem Anwenden von P erstmalig 0.

DENKANSTOSS: $\dim\left(\langle E_1\rangle_A \cap \langle E_{2n}\rangle_A\right) = 2.$

Wir wollen aber aus Dimensionsgründen eines mit Länge 1. Also wiederholen wir den Erfolg:

```
> Test2:=<(P^2).E1|(P^2).A.E1|(P).E2n>;
```

(2.2.13)

$$Test2 := \begin{bmatrix} 1 & -2 & -4 \\ -116 & 100 & 112 \\ 7 & -8 & -12 \\ 8 & -7 & -8 \\ 5 & -4 & -4 \\ 586 & -491 & -528 \\ 0 & 0 & 0 \\ -2 & 1 & 0 \end{bmatrix} \quad (2.2.13)$$

```
> nn:=3*op(NullSpace(Test2));
```

$$nn := \begin{bmatrix} -4 \\ -8 \\ 3 \end{bmatrix} \quad (2.2.14)$$

Wir ersetzen E_{2n} durch (Hier gut aufpassen!):

```
> E2nn:=<P.E1|P.A.E1|E2n>.nn;
```

$$E2nn := \begin{bmatrix} -16 \\ 451 \\ -48 \\ -32 \\ -16 \\ -2130 \\ 0 \\ 0 \end{bmatrix} \quad (2.2.15)$$

```
> P.E2nn;
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix} \quad (2.2.16)$$

Jetzt hat E_{2nn} Länge 1. Immer noch erzeugt es mit E_1 zusammen V, wenn man die Bilder unter A hinzunimmt. Also: Unsere Zauberbasis sind die Spalten von:

```
> T:=Matrix([E1,A.E1,P.E1,A.P.E1,P.P.E1,A.P.P.E1,E2nn,A.E2nn])
  ;
```

$$T := \begin{bmatrix} 1 & 0 & 1 & 0 & 1 & -2 & -16 & 0 \\ 0 & -6 & -21 & -41 & -116 & 100 & 451 & 692 \\ 0 & 2 & 3 & 3 & 7 & -8 & -48 & -32 \\ 0 & 0 & 2 & 3 & 8 & -7 & -32 & -48 \\ 0 & 0 & 0 & 2 & 5 & -4 & -16 & -32 \\ 0 & 23 & 94 & 202 & 586 & -491 & -2130 & -3560 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 3 \\ 0 & 0 & 0 & 0 & -2 & 1 & 0 & 13 \end{bmatrix} \tag{2.2.17}$$

```
> (T^(-1)).A.T,DiagonalMatrix([JBlock(p,x,3),JBlock(p,x,1)]);
```

$$\begin{bmatrix} 0 & -1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & -1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & -1 \end{bmatrix}, \begin{bmatrix} 0 & -1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & -1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & -1 \end{bmatrix} \tag{2.2.18}$$

DENKANSTOSS: Warum haben wir mit (P^2).E1 und (P^2).A.E1 statt mit P.E1 und P.A.E1 gerechnet?

ÜBUNG [05]:

Zerlege den Vektorraum $\mathbb{Q}^{4 \times 1}$ in eine direkte Summe von bezüglich B zyklischen Teilräumen, mit

```
B := Matrix([[-2050,-2965,-3211,-2401],[1152,1344,1152,
576],[0,0,0,0],[4,298,598,706]]);
```

$$B := \begin{bmatrix} -2050 & -2965 & -3211 & -2401 \\ 1152 & 1344 & 1152 & 576 \\ 0 & 0 & 0 & 0 \\ 4 & 298 & 598 & 706 \end{bmatrix}$$

und benutze dies um (wie oben) die Jordan-Normalform herzustellen.

(Hinweis: Wende das eben vorgeführte Verfahren an. Warum ist es in diesem

Fall einfacher? Warum muss man dennoch besonders aufpassen?)

MATH: Wir behandeln an einem Beispiel noch den Fall, dass man eine Summe von mehr als zwei zyklischen Teilräumen hat.

```
> A:=DiagonalMatrix([CompanionMatrix(x^3,x),CompanionMatrix(x^2,x),CompanionMatrix(x,x)]);
```

$$A := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix} \tag{2.2.20}$$

```
> T:=RandomMatrix(6,6,generator=1..9,outputoptions=[shape=triangular[upper,unit]]):
  T:=T.Transpose(T):
> A:=T.A.T^(-1);
```

$$A := \begin{bmatrix} -445 & 1447 & -11359 & 39120 & -84538 & 414167 \\ -722 & 2351 & -18463 & 63589 & -137425 & 673268 \\ -688 & 2105 & -16361 & 56337 & -121469 & 595110 \\ -231 & 711 & -5531 & 19045 & -41072 & 201223 \\ -436 & 1284 & -9916 & 34142 & -73504 & 360120 \\ -84 & 246 & -1898 & 6535 & -14066 & 68914 \end{bmatrix} \tag{2.2.21}$$

```
> MinimalPolynomial(A,x),CharacteristicPolynomial(A,x);
```

$$x^3,\ x^6 \tag{2.2.22}$$

```
> A^2;
```

$$\begin{bmatrix} 103 & -103 & 515 & -1751 & 3296 & -16171 \\ 141 & -141 & 705 & -2397 & 4512 & -22137 \\ 115 & -115 & 575 & -1955 & 3680 & -18055 \\ 46 & -46 & 230 & -782 & 1472 & -7222 \\ 42 & -42 & 210 & -714 & 1344 & -6594 \\ 7 & -7 & 35 & -119 & 224 & -1099 \end{bmatrix} \tag{2.2.23}$$

Jeder Standardbasisvektor hat Länge 3. Unsere Erzeuger sollten aber Länge 3, 2 und 1 haben.

Wir betrachten die ersten drei Basisvektoren:

```
> Rank(<SubMatrix(IdentityMatrix(6),1..6,1..3)|SubMatrix(A,1..6,1..3)|SubMatrix(A^2,1..6,1..3)>);
```

$$6 \tag{2.2.24}$$

Damit erzeugen die drei unter A ganz V. Welchen der drei ersten Standardbasisvektoren werden wir ersetzen?

```
> NullSpace(SubMatrix(A^2,1..6,1..3));
```

$$\left\{ \begin{bmatrix} -5 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 1 \\ 1 \\ 0 \end{bmatrix} \right\} \tag{2.2.25}$$

Also den zweiten und den dritten:

```
> s,t:=op(%);
```

$$s, t := \begin{bmatrix} -5 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 1 \\ 1 \\ 0 \end{bmatrix} \tag{2.2.26}$$

```
> neu1:=SubMatrix(IdentityMatrix(6),1..6,1..3).<s|t>;
```

$$neu1 := \begin{bmatrix} -5 & 1 \\ 0 & 1 \\ 1 & 0 \\ 0 & 0 \\ 0 & 0 \\ 0 & 0 \end{bmatrix} \tag{2.2.27}$$

```
> A.neu1,
  A.A.neu1;
```

$$\begin{bmatrix} -9134 & 1002 \\ -14853 & 1629 \\ -12921 & 1417 \\ -4376 & 480 \\ -7736 & 848 \\ -1478 & 162 \end{bmatrix}, \begin{bmatrix} 0 & 0 \\ 0 & 0 \\ 0 & 0 \\ 0 & 0 \\ 0 & 0 \\ 0 & 0 \end{bmatrix} \tag{2.2.28}$$

Diese beiden haben also noch Länge 2. Wie werden wir weiter ersetzen?

```
> NullSpace(A.<Column(A,1)|neu1>);
```

(2.2.29)

$$\left\{\left[\begin{array}{c} -\frac{4}{283} \\ \frac{31}{283} \\ 1 \end{array}\right]\right\} \tag{2.2.29}$$

Also wird der Dritte ersetzt:

```
> neu2:=<Column(neu1,1)|<Column(A,1)|neu1>.op(%)>;
```

$$neu2 := \left[\begin{array}{cc} -5 & \frac{1908}{283} \\ 0 & \frac{3171}{283} \\ 1 & \frac{2783}{283} \\ 0 & \frac{924}{283} \\ 0 & \frac{1744}{283} \\ 0 & \frac{336}{283} \end{array}\right] \tag{2.2.30}$$

```
> A.neu2;
```

$$\left[\begin{array}{cc} -9134 & 0 \\ -14853 & 0 \\ -12921 & 0 \\ -4376 & 0 \\ -7736 & 0 \\ -1478 & 0 \end{array}\right] \tag{2.2.31}$$

Jetzt ist es gut: Längen 3, 2 ,1 summieren sich zu 6. Man nennt das Längentupel $(3, 2, 1)$ auch die **zugehörige Partition**, die im Detail weiter unten behandelt wird.

DENKANSTOSS: Diese Partition ist dem Problem eindeutig zugeordnet und hängt nicht von der Durchführung des Algorithmus ab.

Die Basis ist

```
> T:=<Column(IdentityMatrix(6),1)|Column(A,1)|Column(A^2,1)
  |Column(neu2,1)|A.Column(neu2,1)|Column(neu2,2)>;
```

$$T := \begin{bmatrix} 1 & -445 & 103 & -5 & -9134 & \frac{1908}{283} \\ 0 & -722 & 141 & 0 & -14853 & \frac{3171}{283} \\ 0 & -688 & 115 & 1 & -12921 & \frac{2783}{283} \\ 0 & -231 & 46 & 0 & -4376 & \frac{924}{283} \\ 0 & -436 & 42 & 0 & -7736 & \frac{1744}{283} \\ 0 & -84 & 7 & 0 & -1478 & \frac{336}{283} \end{bmatrix} \tag{2.2.32}$$

```
> (T^(-1)).A.T,DiagonalMatrix([JBlock(x,x,3),JBlock(x,x,2),
  JBlock(x,x,1)]);
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix} \tag{2.2.33}$$

ÜBUNG [06]:

Führe die Zerlegung am Beispiel von

```
> B:=IdentityMatrix(6)+Transpose(A);
```

$$B := \begin{bmatrix} -444 & -722 & -688 & -231 & -436 & -84 \\ 1447 & 2352 & 2105 & 711 & 1284 & 246 \\ -11359 & -18463 & -16360 & -5531 & -9916 & -1898 \\ 39120 & 63589 & 56337 & 19046 & 34142 & 6535 \\ -84538 & -137425 & -121469 & -41072 & -73503 & -14066 \\ 414167 & 673268 & 595110 & 201223 & 360120 & 68915 \end{bmatrix} \tag{2.2.34}$$

durch.

Die Partition eines Endomorphismus

MATH: Das oben vorgestellte Verfahren ist schnell unübersichtlich, wenn man die Partition des Endomorphismus (also die Länge und Anzahl der Blöcke) nicht bereits kennt. Will man lediglich die Jordan-Normalform ohne die Transformationsmatrix, so reicht schon die Partition alleine aus. Die Partition zu kennen erleichtert allerdings auch das Verfahren zur Bestimmung des Basis. Hier zwei Möglichkeiten zur Bestimmung der Partition am Beipiel von C:

```
> C := Matrix([[-585,-474,10119,10546,-14017,-1617,-6895,872,
  -3631,-688],[-1382,-1132,23602,24795,-32808,-3775,-16114,
  1980,-8495,-1605],[-4029,-3225,69214,72365,-96019,-11064,
  -47198,5906,-24873,-4708],[2785,2154,-48563,-50147,67013,
  7752,33015,-4316,17372,3303],[-12362,-9644,214162,222347,
  -296219,-34207,-145790,18714,-76770,-14569],[-51550,-40704,
  890825,926821,-1233228,-142321,-606739,77277,-319553,
  -60594],[12360,8726,-220343,-223419,301759,35099,149135,
  -20755,78336,14996],[-52629,-45895,881431,941183,-1233773,
  -141252,-604243,69661,-319085,-59917],[187520,158113,
  -3175625,-3359756,4427554,508348,2171935,-259744,1145853,
  215917],[-822332,-701158,13876850,14724709,-19371777,
  -2222151,-9497914,1122822,-5012322,-943439]]);
```

$$C := [[-585, -474, 10119, 10546, -14017, -1617, -6895, 872, -3631, -688],$$
$$[-1382, -1132, 23602, 24795, -32808, -3775, -16114, 1980, -8495, -1605],$$
$$[-4029, -3225, 69214, 72365, -96019, -11064, -47198, 5906, -24873, -4708],$$
$$[2785, 2154, -48563, -50147, 67013, 7752, 33015, -4316, 17372, 3303],$$
$$[-12362, -9644, 214162, 222347, -296219, -34207, -145790, 18714, -76770, -14569],$$
$$[-51550, -40704, 890825, 926821, -1233228, -142321, -606739, 77277, -319553, -60594],$$
$$[12360, 8726, -220343, -223419, 301759, 35099, 149135, -20755, 78336, 14996],$$
$$[-52629, -45895, 881431, 941183, -1233773, -141252, -604243, 69661, -319085, -59917],$$
$$[187520, 158113, -3175625, -3359756, 4427554, 508348, 2171935, -259744, 1145853, 215917],$$
$$[-822332, -701158, 13876850, 14724709, -19371777, -2222151, -9497914, 1122822, -5012322, -943439]] \quad \textbf{(2.3.1)}$$

MATH: Sei $n \in \mathbb{N}$. Ein Tupel $(m_1,\ldots, m_s)$ heißt **Partition** von n, falls für alle i gilt $m_i \neq 0$, $m_i \geq m_{i+1}$ und $\sum_{i=1}^{s} m_i = n$.

Wir ordnen nun einem Endomorphismus eines n-dimensionalen Raumes eine Partition von n zu und nennen sie die **Partition des Endomorphismus**:

```
> factor(MinimalPolynomial(C,x));
  factor(CharacteristicPolynomial(C,x));
```

$$(x-2)^3$$

$$(x-2)^{10} \tag{2.3.2}$$

```
> CC:=C-2:
```

Erste Möglichkeit über **Bilder**:

```
> Rank(IdentityMatrix(10)),Rank(CC),Rank(CC^2),Rank(CC^3);
```

$$10,\ 5,\ 2,\ 0 \tag{2.3.3}$$

```
> (10-5,5-2,2-0);
```

$$5,\ 3,\ 2 \tag{2.3.4}$$

ist die sogenannte **konjugierte Partition**.

Zweite Möglichkeit über **Kerne**:

```
> nops(NullSpace(IdentityMatrix(10))),nops(NullSpace(CC)),nops
  (NullSpace(CC^2)),nops(NullSpace(CC^3));
```

$$0,\ 5,\ 8,\ 10 \tag{2.3.5}$$

```
> (5-0,8-5,10-8);
```

$$5,\ 3,\ 2 \tag{2.3.6}$$

ist wieder die konjugierte Partition.

Wir nehmen nun die konjugierte Partition $(5, 3, 2)$ von 10 und visualisieren die durch

$x\,x\,x\,x\,x$
$x\,x\,x$
$x\,x$

Dieses Schema wird an der Diagonalen gespiegelt (transponiert):

$x\,x\,x$
$x\,x\,x$
$x\,x$
x
x

und wir erhalten die Partition $(3, 3, 2, 1, 1)$ von C. Man beachte: Die Größe des größten Blocks ist immer der Grad des Minimalpolynoms.

MATH: Einem Endomorphismus $\alpha \in End(V)$ ist eine eindeutige Partition, genannt die zu α **gehörige Partition**, zugeordnet, welche durch die Dimension der Räume in einer direkten Zerlegung von V in α-zyklische Räume gegeben ist.

MATH: Sei V ein endlich erzeugter K-Vektorraum und $\alpha \in End(V)$ mit charakteristischem Polynom $\chi_\alpha(x) = p^c$ für ein normiertes irreduzibles Polynom $p \in K[x]$. Ist $a = (a(1),..., a(k)) \in \mathbb{N}^k$ die zu α gehörige Partition von c, so gibt es eine Basis B von V mit ${}^B\alpha^B = Diag(J(p^{a(1)}),..., J(p^{a(k)}))$. Diese Matrix heißt

Jordan-Normalform.

ÜBUNG [07]:

1.) Schreibe mit Hilfe der eben ausgerechneten Partition die Jordan-Normalform von C hin, ohne die zugehörige Basistransformation auszurechnen.
2.) Schreibe alle möglichen Partitionen eines Endomorphismus mit Minimalpolynom p^3 und charakteristischem Polynom p^{10} hin, wobei p ein beliebiges Polynom von Grad 1 ist. (Hinweis: Du darfst dir Hilfe in Maples **combinat**-Paket suchen.) Schreibe für einige von ihnen die zugehörige Jordan-Form für den Fall $p = x - a$ hin.

Jordan-Normalform

```
> interface(rtablesize=20):#damit Maple die nächste Matrix
  auch anzeigen kann
```

MATH: Mit Hilfe der vorgestellten Verfahren lässt sich die Jordan-Normalform einer beliebigen Matrix herstellen. Sie ist eindeutig bis auf die Reihenfolge der Haupträume, die Basis ist es allerdings nicht.

ÜBUNG [08]:

Bestimme eine Basis, die die folgende Matrix E über $\mathbb{Q}$ auf Jordan-Normalform transformiert, sowie die zugehörige Jordan-Normalform.

```
> E:=Matrix([[275021318662427290940,-921450713945492110586,
  2453606792486202458837,-3837810574795276766056,
  4001130293392669817196,-5087428337352354046535,
  1441145746614411874471,2082203903468764335393,
  -4941456758923748098857,3494002585770741408845,
  -2763658307219555998694,3136403345001268019758,
  -4614177252374464554244,-5573936352142738880160,
  -1740530619726901686759],[54735511096680626937,
  -183389695109836703838,488323677852276332288,
  -763811781302332759419,796316127921973910781,
  -1012514199142919431122,286820852250257982112,
  414406037387081298268,-983462527840717281338,
  695386154918515629660,-550030996425862264857,
  624215755087340270021,-918326446210505045987,
  -1109340383279099019042,-346405266011800253538],
  [-5275394552374491653,17675051884236843351,
  -47064511106640583012,73615984018209735040,
  -76748744627366552161,97585859405345025315,
  -27643720340818731740,-39940347834542928648,
  94785866759724034134,-67021139657838133863,
  53011846679514659677,-60161754735013038917,
```

88508067880891965833,106917942254027654033,
33386450891171265778],[2722077124931574669,
-9120238105110811998,24285051631220596041,
-37985478457739415141,39601967216525079856,
-50353813912271975978,14264021020736461810,
20609019121287091957,-48909031753335464537,
34582571850920528406,-27353846951928033916,
31043163640947221418,-45669719024527696035,
-55169121847662584254,-17227241176226856157],
[-528799482172912163,1771726871036938419,
-4717692459744254515,7379181564904368022,
-7693205885066025699,9781894303567892888,
-2770974731165500059,-4003574527559439064,
9501226261321512649,-6718121951608589991,
5313846536945706452,-6030545096606906002,
8871946922429627232,10717331554561191957,
3346619436246007149],[-56716525441546926,
190027024504135134,-505997326848606973,791456030259155650,
-825136790063432418,1049159607447197330,
-297201612588979076,-429404423045922801,
1019056483871269229,-720553910193089963,569937986827571811,
-646807676498875623,951562965368855501,1149490172119634449,
358942534548367988],[6829498508865867,-22882030773060237,
60929472711965092,-95302872247713576,99358527933107882,
-126334148977606832,35787417409820951,51706567778311656,
-122709292977182574,86765220840223845,-68628862591427668,
77885096588431182,-114582087010640512,-138415415178112556,
-43221926685093820],[1685737437363180,-5648012934925324,
15039331666203395,-23523779883308889,24524844692503489,
-31183285899000504,8833472799747367,12762825403382269,
-30288556153100771,21416415984421111,-16939785959843123,
19224518895633400,-28282503242889455,-34165326629829128,
-10668546136044475],[-570726205014827,1912201103665996,
-5091742342330493,7964251918987695,-8303174165646742,
10557467626088858,-2990675947740913,-4321004414036086,
10254546363840064,-7250779124286824,5735163460459944,
-6508686625443272,9575373594033308,11567072533601344,
3611961574981558],[283081090861199,-948454739968805,
2525512170274536,-3950281415701150,4118387380455265,
-5236520464747078,1483379950427389,2143224951641034,
-5086271044617423,3596397792967601,-2844650051808365,
3228318752178110,-4749400287903652,-5737286078890674,
-1791538593833784],[-21007977229181,70386600247219,
-187422981892585,293157772488464,-305633230561086,
388611978106093,-110084400645203,-159052732361623,
377461687614661,-266895406942102,211106800999043,
-239579572866461,352461878674362,425774730969122,
132953429952770],[-11562714168080,38740528469704,

```
-103156926741552,161352970464701,-168219417157065,
213890617648291,-60590053248774,-87542044712814,
207753538360141,-146898260097180,116192414534242,
-131863724492416,193993734553842,234344861495517,
73177083705453],[3641290229729,-12200034157986,
32485825042999,-50812723237425,52975081045115,
-67357698629971,19080811452471,27568440028181,
-65425030701240,46260695498690,-36590916083979,
41526071186902,-61091840549379,-73799078845857,
-23044675849098],[337363329312,-1130325758619,
3009791969445,-4707767961695,4908109099879,-6240649888045,
1767825597898,2554199233000,-6061589377696,4286025355034,
-3390126162356,3847365286742,-5660121940604,-6837439854769,
-2135075227981],[33396666821,-111894534700,297948860720,
-466036893853,485869299046,-617781741388,175002667344,
252848289565,-600055973118,424287254545,-335599349746,
380862901982,-560313437195,-676859874525,-211357873947]]):
Dimension(E);
```

15, 15